

# ZScript Opcodes Reference

Zenotta AG

To express redeeming conditions for transactions, Zenotta uses ZScript, a stack-based scripting language. We provide a list of the opcodes that ZScript currently supports, along with examples. New opcodes will be added with future updates.

## Constants

OP\_0 ... OP\_16 push numbers from 0 to 16 onto the stack. Negative numbers are not allowed.

```
OP_1 OP_2 OP_3 -> [1 2 3]
```

## Flow Control

OP\_NOP does nothing.

```
OP_0 OP_NOP -> [0]
```

OP\_IF checks if the top item on the stack is not 0 and executes the next block of instructions until OP\_ENDIF or OP\_ELSE is encountered. Skips the next block if the condition check is failed. Should have a closing OP\_ENDIF. The top item is consumed.

```
OP_1 OP_IF OP_2 OP_ELSE OP_3 OP_ENDIF -> [2]
```

OP\_NOTIF checks if the top item on the stack is 0 and executes the next block of instructions until OP\_ENDIF or OP\_ELSE is encountered. Skips the next block if the condition check is failed. Should have a closing OP\_ENDIF. The top item is consumed.

```
OP_0 OP_NOTIF OP_2 OP_ELSE OP_3 OP_ENDIF -> [2]
```

OP\_ELSE executes the next block of instructions if the previous OP\_IF or OP\_NOTIF was not executed. Should have a closing OP\_ENDIF.

```
OP_0 OP_IF OP_2 OP_ELSE OP_3 OP_ENDIF -> [3]
```

OP\_ENDIF ends an OP\_IF or OP\_NOTIF block.

```
OP_0 OP_IF OP_2 OP_ENDIF -> []
```

OP\_VERIFY removes the top item from the stack and ends execution with an error if it is 0.

```
OP_1 OP_VERIFY -> []
```

```
OP_0 OP_VERIFY -> fail
```

OP\_RETURN Ends execution with an error.

```
OP_RETURN -> fail
```

## Stack

OP\_TOALTSTACK moves the top item from the main stack to the top of the alt stack.

```
OP_1 OP_TOALTSTACK -> [] [1]
```

OP\_FROMALTSTACK moves the top item from the alt stack to the top of the main stack.

```
OP_1 OP_TOALTSTACK OP_FROMALTSTACK -> [1] []
```

OP\_2DROP removes the top two items from the stack.

```
OP_1 OP_2 OP_2DROP -> []
```

OP\_2DUP duplicates the top two items on the stack.

```
OP_1 OP_2 OP_2DUP -> [1 2 1 2]
```

OP\_3DUP duplicates the top three items on the stack.

```
OP_1 OP_2 OP_3 OP_3DUP -> [1 2 3 1 2 3]
```

OP\_2OVER copies the second-to-top pair of items to the top of the stack.

```
OP_1 OP_2 OP_3 OP_4 OP_2OVER -> [1 2 3 4 1 2]
```

OP\_2ROT moves the third-to-top pair of items to the top of the stack.

```
OP_1 OP_2 OP_3 OP_4 OP_5 OP_6 OP_2ROT -> [3 4 5 6 1 2]
```

OP\_2SWAP swaps the top two pairs of items on the stack.

```
OP_1 OP_2 OP_3 OP_4 OP_2SWAP -> [3 4 1 2]
```

OP\_IFDUP duplicates the top item on the stack if it is not 0.

```
OP_1 OP_IFDUP -> [1 1]  
OP_0 OP_IFDUP -> [0]
```

OP\_DEPTH pushes the stack size onto the stack.

```
OP_0 OP_1 OP_2 OP_3 OP_DEPTH -> [0 1 2 3 4]
```

OP\_DROP removes the top item from the stack.

```
OP_1 OP_DROP -> []
```

OP\_DUP duplicates the top item on the stack.

```
OP_1 OP_DUP -> [1 1]
```

OP\_NIP removes the second-to-top item from the stack.

```
OP_1 OP_2 OP_NIP -> [2]
```

OP\_OVER copies the second-to-top item to the top of the stack.

```
OP_1 OP_2 OP_OVER -> [1 2 1]
```

OP\_PICK copies the  $(n + 1)$ th-to-top item to the top of the stack, where  $n$  is the top item on the stack.

```
OP_1 OP_2 OP_3 OP_4 OP_3 OP_PICK -> [1 2 3 4 1]
```

OP\_ROLL moves the  $(n + 1)$ th-to-top item to the top of the stack, where  $n$  is the top item on the stack.

```
OP_1 OP_2 OP_3 OP_4 OP_3 OP_ROLL -> [2 3 4 1]
```

OP\_ROT moves the third-to-top item to the top of the stack.

```
OP_1 OP_2 OP_3 OP_ROT -> [2 3 1]
```

OP\_SWAP swaps the top two items on the stack.

```
OP_1 OP_2 OP_SWAP -> [2 1]
```

OP\_TUCK copies the top item behind the second-to-top item on the stack.

```
OP_1 OP_2 OP_TUCK -> [2 1 2]
```

## Splice

OP\_CAT concatenates the two strings on top of the stack.

```
"hello" "world" OP_CAT -> ["helloworld"]
```

OP\_SUBSTR extracts a substring from the third-to-top item on the stack.

```
"hello" 1 2 OP_SUBSTR -> ["e1"]
```

OP\_LEFT extracts a left substring from the second-to-top item on the stack.

```
"hello" 2 OP_LEFT -> ["he"]
```

OP\_RIGHT extracts a right substring from the second-to-top item on the stack.

```
"hello" 2 OP_RIGHT -> ["llo"]
```

OP\_SIZE computes the size in bytes of the string on top of the stack.

```
"hello" OP_SIZE -> ["hello" 5]
```

## Bitwise Logic

OP\_INVERT computes bitwise NOT of the number on top of the stack.

```
OP_0 OP_INVERT -> [18446744073709551615] // =  $2^{64} - 1$ 
```

OP\_AND computes bitwise AND between the two numbers on top of the stack.

```
OP_1 OP_2 OP_AND -> [0]
```

OP\_OR computes bitwise OR between the two numbers on top of the stack.

```
OP_1 OP_1 OP_OR -> [1]
```

OP\_XOR computes bitwise XOR between the two numbers on top of the stack.

```
OP_1 OP_1 OP_XOR -> [0]
```

OP\_EQUAL substitutes the top two items on the stack with 1 if they are equal, with 0 otherwise.

```
OP_1 OP_1 OP_EQUAL -> [1]
```

```
OP_1 OP_2 OP_EQUAL -> [0]
```

OP\_EQUALVERIFY computes OP\_EQUAL and OP\_VERIFY in sequence.

```
OP_1 OP_1 OP_EQUALVERIFY -> []
```

```
OP_1 OP_2 OP_EQUALVERIFY -> fail
```

## Arithmetic

OP\_1ADD adds 1 to the number on top of the stack. Ends execution with an error in case of overflow.

```
OP_1 OP_1ADD -> [2]
OP_0 OP_INVERT OP_1ADD -> fail // = (2^64 - 1) + 1
```

OP\_1SUB subtracts 1 from the number on top of the stack. Ends execution with an error in case the result is negative.

```
OP_1 OP_1SUB -> [0]
OP_0 OP_1SUB -> fail // = -1
```

OP\_2MUL multiplies by 2 the number on top of the stack. Ends execution with an error in case of overflow.

```
OP_2 OP_2MUL -> [4]
OP_0 OP_INVERT OP_2MUL -> fail // = (2^64 - 1) * 2
```

OP\_2DIV divides by 2 the number on top of the stack.

```
OP_1 OP_2DIV -> [0]
```

OP\_NOT substitutes the number on top of the stack with 1 if it is equal to 0, with 0 otherwise.

```
OP_0 OP_NOT -> [1]
OP_2 OP_NOT -> [0]
```

OP\_ONOTEQUAL substitutes the number on top of the stack with 1 if it is not equal to 0, with 0 otherwise.

```
OP_2 OP_ONOTEQUAL -> [1]
OP_0 OP_ONOTEQUAL -> [0]
```

OP\_ADD adds the two numbers on top of the stack. Ends execution with an error in case of overflow.

```
OP_3 OP_2 OP_ADD -> [5]
OP_0 OP_INVERT OP_2 OP_ADD -> fail // = (2^64 - 1) + 2
```

OP\_SUB subtracts the number on top of the stack from the second-to-top number on the stack. Ends execution with an error in case the result is negative.

```
OP_3 OP_2 OP_SUB -> [1]
OP_2 OP_3 OP_SUB -> fail // = -1
```

OP\_MUL multiplies the second-to-top number by the number on top of the stack. Ends execution with an error in case of overflow.

```
OP_3 OP_2 OP_MUL -> [6]
OP_0 OP_INVERT OP_3 OP_MUL -> fail // = (2^64 - 1) * 3
```

OP\_DIV divides the second-to-top number by the number on top of the stack. Ends execution with an error in case of division by 0.

```
OP_3 OP_2 OP_DIV -> [1]
OP_3 OP_0 OP_DIV -> fail
```

OP\_MOD computes the remainder of the division of the second-to-top number by the number on top of the stack. Ends execution with an error in case of division by 0.

```
OP_3 OP_2 OP_MOD -> [1]
OP_3 OP_0 OP_MOD -> fail
```

OP\_LSHIFT computes the left shift of the second-to-top number by the number on top of the stack. Ends execution with an error in case of overflow.

```
OP_2 OP_1 OP_LSHIFT -> [4]
OP_1 <64> OP_LSHIFT -> fail
```

OP\_RSHIFT computes the right shift of the second-to-top number by the number on top of the stack. Ends execution with an error in case of overflow.

```
OP_2 OP_1 OP_RSHIFT -> [1]
OP_1 <64> OP_RSHIFT -> fail
```

OP\_BOOLAND substitutes the two numbers on top of the stack with 1 if they are both non-0, with 0 otherwise.

```
OP_1 OP_2 OP_BOOLAND -> [1]
OP_0 OP_1 OP_BOOLAND -> [0]
```

OP\_BOOLOR substitutes the two numbers on top of the stack with 1 if they are not both 0, with 0 otherwise.

```
OP_0 OP_1 OP_BOOLOR -> [1]
OP_0 OP_0 OP_BOOLOR -> [0]
```

OP\_NUMEQUAL substitutes the two numbers on top of the stack with 1 if they are equal, with 0 otherwise.

```
OP_1 OP_1 OP_NUMEQUAL -> [1]
OP_0 OP_1 OP_NUMEQUAL -> [0]
```

OP\_NUMEQUALVERIFY computes OP\_NUMEQUAL and OP\_VERIFY in sequence.

```
OP_1 OP_1 OP_NUMEQUALVERIFY -> []
OP_0 OP_1 OP_NUMEQUALVERIFY -> fail
```

OP\_NUMNOTEQUAL substitutes the two numbers on top of the stack with 1 if they are not equal, with 0 otherwise.

```
OP_0 OP_1 OP_NUMNOTEQUAL -> [1]
OP_1 OP_1 OP_NUMNOTEQUAL -> [0]
```

OP\_LESSTHAN substitutes the two numbers on top of the stack with 1 if the second-to-top is less than the top item, with 0 otherwise.

```
OP_2 OP_3 OP_LESSTHAN -> [1]
OP_3 OP_2 OP_LESSTHAN -> [0]
```

OP\_GREATERTHAN substitutes the two numbers on top of the stack with 1 if the second-to-top is greater than the top item, with 0 otherwise.

```
OP_3 OP_2 OP_GREATERTHAN -> [1]
OP_2 OP_3 OP_GREATERTHAN -> [0]
```

OP\_LESSTHANOREQUAL substitutes the two numbers on top of the stack with 1 if the second-to-top is less than or equal to the top item, with 0 otherwise.

```
OP_2 OP_2 OP_LESSTHANOREQUAL -> [1]
OP_3 OP_2 OP_LESSTHANOREQUAL -> [0]
```

OP\_GREATERTHANOREQUAL substitutes the two numbers on top of the stack with 1 if the second-to-top is greater than or equal to the top item, with 0 otherwise.

```
OP_2 OP_2 OP_GREATERTHANOREQUAL -> [1]
OP_2 OP_3 OP_GREATERTHANOREQUAL -> [0]
```

OP\_MIN substitutes the two numbers on top of the stack with the minimum between the two.

```
OP_0 OP_1 OP_MIN -> [0]
```

OP\_MAX substitutes the two numbers on top of the stack with the maximum between the two.

```
OP_0 OP_1 OP_MAX -> [1]
```

OP\_WITHIN substitutes the three numbers on top of the stack with 1 if the third-to-top is greater or equal to the second-to-top and less than the top item, with 0 otherwise.

```
OP_1 OP_1 OP_3 OP_WITHIN -> [1]
OP_3 OP_1 OP_3 OP_WITHIN -> [0]
```

## Crypto

OP\_SHA3 hashes the top item on the stack using SHA3-256.

```
"hello" OP_SHA3 -> ["3338be694f50c5f338814986cdf0686453a888b84f424d792af4b9202398f392"]
```

OP\_HASH256 creates a standard address from a public key and pushes it onto the stack.

```
<pk> OP_HASH256 -> ["3c85fab8f52253e477abf4933cb196622d4cd57eb57fb236db6026017e576c9f"]
```

OP\_CHECKSIG pushes 1 onto the stack if the signature is valid, 0 otherwise. It allows signature verification on arbitrary messages, not only transactions.

```
<msg> <sig> <pk> OP_CHECKSIG -> [1] / [0]
```

OP\_CHECKSIGVERIFY runs OP\_CHECKSIG and OP\_VERIFY in sequence.

```
<msg> <sig> <pk> OP_CHECKSIGVERIFY -> [] / fail
```

OP\_CHECKMULTISIG pushes 1 onto the stack if the  $m$ -of- $n$  multi-signature is valid, 0 otherwise. It allows multi-signature verification on arbitrary messages, not only transactions. Ordering of signatures and public keys is not relevant.

```
<msg> <sig1> <sig2> OP_2 <pk1> <pk2> <pk3> OP_3 OP_CHECKMULTISIG -> [1] / [0]
```

OP\_CHECKMULTISIGVERIFY runs OP\_CHECKMULTISIG and OP\_VERIFY in sequence.

```
<msg> <sig1> <sig2> OP_2 <pk1> <pk2> <pk3> OP_3 OP_CHECKMULTISIGVERIFY -> [] / fail
```